

Mean and Individual Subject Graphs of Concentration vs. Time Data Using PROC SGPLOT

Pooja Trivedi, Cadila Healthcare Limited, Ahmedabad, India

ABSTRACT

In clinical research Industry, Clinical Study Report (CSR) is very important document because it contains all the trial related activities, results and conclusion of trial. Graphs are also one of the important parts of the CSR as graphs are visual representation of the data and help us to understand the data in better way. The main purpose of this paper is to create highly presentable graphs which can be directly incorporated into CSR or as appendices. The mean and individual subject's graphs of concentration vs. time data are most common graphs which show change in drug concentration in the body with respect to time.

INTRODUCTION

The mean and individual subject's graphs of concentration vs. time data are most widely used in the Pharmacokinetic world as they play an important role in showing the drug's concentration profiles. Pharmacokinetic (PK) may be defined as the absorption, distribution, metabolism, and excretion of an investigational drug. In a new drug development process, studies such as PK studies are conducted to examine what is happening to the drug within the body. These studies involve analyzing drug concentrations in samples collected from the blood, urine, feces and other matrices at different time intervals. Graphs play an important role in interpreting drug concentration-time data as they provide a quick view of the general trend of the drug concentrations over time. In general, this graph is very effective to show the mean concentration-time curves for a dataset with enough quantifiable concentrations over time.

Usually presenting the drug's concentration is done with a line graph. This paper outlines the simplest, easiest, helpful and presentable methods of plotting graphs by using PROC SGPLOT which reduces time and efforts required to prepare graphs. This paper highlights the SGPLOT procedure with SERIES Statement and rich text file (.rtf) as output delivery destination to produce linear and semi-log graphs of mean concentration vs. time on single page and also to produce the same for Individual subjects considering one subject on one page.

Below mentioned topics are discussed in detail:

1. PROC SGPLOT for overall mean concentration.
2. An approach to generate two different graphs on one page using **STARTPAGE=NO** option.
3. Avoid graphics title to display using **NOGTITLE** option
4. Write title in word's body part using **BODYTITLE** option

MEAN PLOT USING PROC SGPLOT

There are four variables in the example data set CP - Subject Concentration Data by Actual Time. Subject identifier as Subject, Treatment arm as Trt, Sampling time as Time, Concentration of drug at that time as Conc are entered for illustrative purpose. The SAS code for creating this data set is shown below:

```
data cp;
  input Subject Trt $ Time Conc ;
datalines;
1 R 0.00 0
1 R 0.25 22.504
1 R 0.52 98.228
1 R 0.75 162.002
1 R 1.00 224.030
1 R 1.25 289.837
```

```

.....
1   R    14.00  62.267
1   R    24.00  36.661
1   R    49.17  12.476
1   R    72.20  7.234
1   R    96.50  4.112
1   R   120.12  2.849
2   T     0.00   0
2   T     0.25  21.218
2   T     0.52  100.025
2   T     0.75  160.666
.....
2   T    72.20  6.523
2   T    97.22  4.080
2   T   120.13  2.875
3   T     0.00   0
3   T     0.25  20.291
3   T     0.50  96.217
3   T     0.75  133.545
3   T     1.00  200.607
.....
3   T    48.67  10.691
3   T    73.60  5.783
3   T    96.60  3.829
3   T   120.38  2.414
;
run;

```

To prepare the mean graph, firstly the mean is calculated using PROC MEANS and then that output dataset is used in PROC SGPLOT to create a mean graph of concentration vs. time data.

```

proc sort data=cp;
  by Trt Time;
run;

proc means data=cp noprint;
  var Conc;
  by Trt Time;
  output out=cp1 (drop= _TYPE_ _FREQ_) mean=mean;
run;

```

In this paper, mean graph has been presented as linear plot as well as semi-log plot on single page by using **STARTPAGE=NO** option in ods rtf statement.

The following SAS codes have been used to generate desired output:

```

***** Set page orientation, size and margins *****;

options nonumber nodate orientation=portrait papersize=letter
      topmargin="0.5in"  bottommargin="0.3in"
      leftmargin="1in"   rightmargin="1in";

***** Assign appropriate path in ods rtf statement to obtain output file
at that location *****;

ods rtf file=" D:\Documents\PharmaSUG\Mean graph.rtf" startpage=no nogtitle
      bodytitle;

```

```
ods escapechar='^' ;
***** Assign graph's size by using height and width option in ods
graphics statement *****;

ods graphics / reset=all border=off height=3in width=6.5in;

data test;
  length value LineColor markercolor MarkerSymbol $30
  LineThickness LinePattern markersize 6. ;
  Id='X'; Value="T"; LineColor='Blue'; LinePattern = 2; LineThickness = 1 ;
  markercolor= 'Blue'; MarkerSymbol= 'triangle'; markersize = 9; output;
  Id='X'; Value="R"; LineColor='Red'; LinePattern = 1; LineThickness = 1 ;
  markercolor= 'Red'; MarkerSymbol= 'circle'; markersize = 8; output;
run;

title1 justify=left HEIGHT=12pt FONT='Times New Roman' bold "Figures A" ;
title2 justify=left HEIGHT=12pt FONT='Times New Roman' bold "Mean Plasma
Concentration vs. Time curve of Drug A (n = &max)" ;
```

LINEAR PLOT

Following SAS code is used to generate the Linear Plot:

```
title4 justify=left HEIGHT=12pt FONT='Times New Roman'
"^S={font_weight=bold}Figure 1: ^S={font_weight=light} Linear Plot of Mean
Plasma Concentrations versus Time Curves of Drug A to healthy, adult, male
human subjects under fasting condition." ;

proc sgplot data=cpl dattrmap=test noborder ;
  series x = time y = mean /group=Trt attrid=X markers;
  xaxis label = 'Time (hr)' values = (0 to 140 by 20) minor MINORCOUNT=1;
  yaxis label = 'Mean (ng/mL)' values = (0 to 350 by 50) minor
  MINORCOUNT=1;
  keylegend / title="Mean Vs. Time" location=inside position=TOPRIGHT
  across=1 down=2 noborder;
run;
```

SEMI-LOG PLOT

Following SAS code is used to generate the Semi-log Plot:

```
title1 justify=left HEIGHT=12pt FONT='Times New Roman'
"^S={font_weight=bold}Figure 2: ^S={font_weight=light}Semi-logarithmic Plot
of Mean Plasma Concentrations versus Time Curves of Drug A to healthy,
adult, male human subjects under fasting condition." ;

proc sgplot data=cpl dattrmap=test noborder ;
  series x = time y = mean /group=Trt attrid=X markers;
  xaxis label = 'Time (hr)' values = (0 to 140 by 20) minor MINORCOUNT=1;
  yaxis label = 'Mean (ng/mL)' values = (1 10 100 1000) TYPE=LOG logbase=10
  LOGSTYLE=LOGEXPAND minor MINORCOUNT=8;
  keylegend / title="Mean Vs. Time" location=inside position=TOPRIGHT
  across=1 down=2 noborder;
run;

ods rtf close;
```

Following figures have been generated by using above codes.

Figures A

Mean Plasma Concentration vs. Time curve of Drug A (n = 3)

Figure 1: Linear Plot of Mean Plasma Concentrations versus Time Curves of Drug A to healthy, adult, male human subjects under fasting condition.





Figure 2: Semi-logarithmic Plot of Mean Plasma Concentrations versus Time Curves of Drug A to healthy, adult, male human subjects under fasting condition.



INDIVIDUAL SUBJECT PLOTS USING PROC SGPLOT

For Individual graphs of each subject, use original dataset CP. Here Dataset “count” has been generated to create variables count1 and count2 and later on their values are used to generate macro variables. These macro variables are further used in title statement to assign different figure number increasing by 1 i.e. Figure 1, Figure 2, Figure 3 and so on till last figure of last subject.

```
proc sort data=CP;
  by subject;
run;

data count;
  set CP;
  by subject;
  retain count1 count2;
  length Trt1 $15. ;
  Trt1 = ifc (Trt = 'R', 'Reference (R)', 'Test (T)');
  if subject = 1 then do;
    count1 = 1;
    count2 = count1+1;
  end;
  else if subject ne 1 and first.subject then do;
    count1 = count2+1;
    count2 = count1+1;
  end;
run;

***** Set page orientation, size and margins *****;

options mprint mlogic symbolgen nonumber nodate orientation=portrait
papersize=letter
  topmargin="0.5in"
  bottommargin="0.3in"
  leftmargin="1in"
  rightmargin="1in";
ods escapechar='^' ;

title1 justify=left HEIGHT=12pt FONT='Times New Roman' bold "Figures B" ;
title2 justify=left HEIGHT=12pt FONT='Times New Roman' bold "Individual
Plasma Concentration vs. Time curves of Drug A (n = &max)" ;
footnotel " ";

ods rtf file=" D:\Documents\PharmaSUG\Individual graph.rtf" startpage=no
nogtitle bodytitle;
```

Both Linear and semi-logarithmic plots are generated inside macro named as “graph”.

This macro generates two graphs; one is linear plot and the other is semi-log plot for one subject (e.g. subject no. 1) on single page. The same process occurs with all subjects. Here in this paper, I have presented graph for subject no. 1 for illustration.

The following SAS codes have been used to generate desired output:

```

%macro graph;

%do i=1 %to &max;

data _null_;
  set count;
  by Subject;
  where Subject = &i ;
  call symputx ("Subid&i",Subject);
  call symputx ("Trt&i",Trt1);
  ***** For Title of figure no. only *****;
  call symputx ("count1&i",count1);
  call symputx ("count2&i",count2);
run;

ods graphics / reset=all border=off height=3in width=6.5in;

data test;
  length value LineColor markercolor MarkerSymbol $30
  LineThickness LinePattern markersize 6. ;
  Id='X'; Value="T"; LineColor='Blue'; LinePattern = 2; LineThickness = 1;
  markercolor= 'Blue'; MarkerSymbol= 'triangle'; markersize = 9; output;
  Id='X'; Value="R"; LineColor='Red'; LinePattern = 1; LineThickness = 1;
  markercolor= 'Red'; MarkerSymbol= 'circle'; markersize = 8; output;
run;

***** 1: LINEAR GRAPH *****;

proc sgplot data=cp (WHERE= (Subject=&i)) dattrmap=test noborder ;
title4 justify=left HEIGHT=12pt FONT='Times New Roman'
"^S={font_weight=bold}Figure " "&count1&i" " : ^S={font_weight=light}
Linear Plot of Individual Plasma Concentrations versus Time curves of";
title5 justify=left HEIGHT=12pt FONT='Times New Roman' "Drug A after
administration of &Trt&i formulation to healthy, adult, male, human
subjects under fasting condition." ;
title7 justify=center HEIGHT=10pt FONT='Times New Roman' "Subject No.=
&&Subid&i" ;
  series x = time y = conc
  /group=Trt attrid=X markers;
  xaxis label = 'Time (hr)' values = (0 to 140 by 20) minor MINORCOUNT=1;
  yaxis label = 'Concentration (ng/mL)' values = (0 to 350 by 50) minor
  MINORCOUNT=1;
  keylegend / title="Concentration Vs. Time" location=inside
  position=TOPRIGHT across=1 down=2 noborder;
run;

***** 2: SEMILOG GRAPH *****;

proc sgplot data=cp (WHERE= (Subject=&i)) dattrmap=test noborder ;
title1 justify=left HEIGHT=12pt FONT='Times New Roman'
"^S={font_weight=bold}Figure " "&count2&i" " : ^S={font_weight=light} Semi-
logarithmic of Individual Plasma Concentrations versus Time curves of";
title2 justify=left HEIGHT=12pt FONT='Times New Roman' "Drug A after
administration of &Trt&i formulation to healthy, adult, male human
subjects under fasting condition." ;

```

```
title4 justify=center HEIGHT=10pt FONT='Times New Roman' "Subject No.=
&&Subid&i" ;
  series x = time y = conc
    /group=Trt attrid=X markers;
  xaxis label = 'Time (hr)' values = (0 to 140 by 20) minor MINORCOUNT=1;
  yaxis TYPE=LOG label = 'Concentration (ng/mL)' values = (1 10 100 1000)
    LOGBASE=10 logstyle=LOGEXPAND minor MINORCOUNT=8;
  keylegend / title="Concentration Vs. Time" location=inside
    position=TOPRIGHT across=1 down=2 noborder;
run;
title;
footnote;

%end;
%mend graph;
%graph

ods rtf close;
```

In this macro definition, the loop gets iterated using do-end loop till last subject of the dataset. This procedure is followed to create graphs for each subject.

Firstly, assign variable's values to create macro variables which can further be used for various purposes.

Create test dataset and map it by using **DATTRMAP=TEST** option in the PROC SGPLOT statement. Further, use **ATTRID=X** option in **SERIES** statement to map test dataset by ID variable. It controls visual attributes of the graph like linecolor, linepattern, markercolor, markersymbol, etc.

After that write SAS code to generate graphs using PROC SGPLOT. In that procedure, use **SERIES** statement with **GROUP=TRT** option. So, that graph can be generated for different values of treatment.

After executing above SAS codes, following outputs have been generated in .rtf file. All texts (i.e. before or between graphs) are part of output generated using SAS code.

Figures B
Individual Plasma Concentration vs. Time curves of Drug A (n = 3)

Figure 1: Linear Plot of Individual Plasma Concentrations versus Time curves of Drug A after administration of Reference (R) formulation to healthy, adult, male, human subjects under fasting condition.





Figure 2: Semi-logarithmic of Individual Plasma Concentrations versus Time curves of Drug A after administration of Reference (R) formulation to healthy, adult, male human subjects under fasting condition.



CONCLUSION

This paper mainly focuses on steps to neatly generate graphs by using PROC SGPLOT with other options and .rtf as an ods destination.

This paper is helpful as it gives answers to many questions like how to generate two graphs on one page using **STATRTPAGE=NO** option in ods rtf statement, how to avoid graphics title using **NOGTITLE** option and it also illustrate how to write title in word body part using **BODYTITLE** option. The flexibility that is now available gives user the ability to customize outputs as per requirement. Minimal SAS code is now needed to create outputs as compared to when SG procedures did not exist.

REFERENCES

Katrina E. Canonizado and Matthew Murphy. "Creating Pharmacokinetic Graphs using SAS/GRAPH®". *Proceedings of the SAS Global Forum 2012*.

Available at <http://support.sas.com/resources/papers/proceedings12/203-2012.pdf>.

Jenny Zhang. "Plotting Overall Mean and Individual Values in One Graph ". *Proceedings of the NESUG 2008*. Available at <http://www.lexjansen.com/nesug/nesug08/cc/cc20.pdf>.

ACKNOWLEDGMENTS

I would like to thank my family for their support, and encouragement and the almighty for everything.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Pooja Trivedi

pctrivedi11@gmail.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.